

Math 420/620 Quiz 8

1. Reconsider the war problem of Example 6.1. In this problem we explore the effects of weather on combat. Bad weather and poor visibility decrease the effectiveness of direct fire weapons for both sides. The effectiveness of indirect fire weapons is relatively unaffected by the weather. Represent the effects of bad weather in the model as follows. Let w denote the decrease in weapon effectiveness caused by bad weather conditions. Thus,

$$R_{n+1} = R_n - w\lambda(0.05)B_n - \lambda(0.005)R_nB_n$$

$$B_{n+1} = B_n - w0.05R_n - 0.005R_nB_n.$$

Here $0 \leq w \leq 1$ with $w = 1$ indicating the best weather and $w = 0$ indicating the worst weather.

- (i) Use a computer to simulate the discrete-time dynamical system given above in the case $\lambda = 3$. Assume that adverse weather conditions cause a 75% decrease in weapon effectiveness for both sides ($w = 0.25$). Who wins the battle, and how long does it take? How many divisions of troops remain on the winning side?

In Example 6.1 it was assumed that $R_0 = 5$ and $B_0 = 2$.

```
In [1]: R0=5; B0=2;
```

```
In [2]: function battle(lambda,w)
    dR(R,B)=-w*lambda*0.05*B-lambda*0.005*R*B
    dB(R,B)=-w*0.05*R-0.005*R*B
    Rn=R0; Bn=B0
    nmax=1000
    for n=1:nmax
        Rn=Rn+dR(Rn,Bn)
        Bn=Bn+dB(Rn,Bn)
        if Rn<=0 && Bn<=0
            return n,"tie",max(Rn,Bn)
        elseif Rn<=0
            return n,"blue",Bn
        elseif Bn<=0
            return n,"red",Rn
        end
    end
    return nmax,"tie",max(Rn,Bn)
end
```

```
Out[2]: battle (generic function with 1 method)
```

```
In [3]: battle(3,0.25)
```

Out[3]: (41, "red", 2.1198506097885152)

The red side wins the battle. It took $n = 41$ hours.

There were 2.1198506097885152 red troops remaining.

(ii) Repeat your analysis for $w = 0.1, 0.2, 0.5, 0.75, 0.9$ and tabulate your results. Answer the same questions as in part (i).

```
In [4]: ws=[0.1,0.2,0.25,0.5,0.75,0.9]
t1=[]
for w in ws
  push!(t1,[(w,battle(3,w)...)...;])
end
T1=reduce(vcat,t1)
```

```
Out[4]: 6×4 Matrix{Any}:
 0.1  110  "red"  1.22379
 0.2   52  "red"  1.92336
 0.25  41  "red"  2.11985
 0.5   21  "red"  2.59439
 0.75  14  "red"  2.76393
 0.9   12  "red"  2.81057
```

```
In [5]: using PrettyTables
pretty_table(T1,column_labels=["w","Hours","Winner","Remaining"])
```

w	Hours	Winner	Remaining
0.1	110	red	1.22379
0.2	52	red	1.92336
0.25	41	red	2.11985
0.5	21	red	2.59439
0.75	14	red	2.76393
0.9	12	red	2.81057

(iii) Which side benefits from fighting in adverse weather conditions? If you were the blue commander, would you expect red to attack on a sunny day or a rainy day?

Recall that small values of w represent adverse weather conditions. As w decreases red wins but has fewer troops remaining. Even though blue loses in all cases, they benefit from adverse weather conditions. Therefore, if I were the the blue commander, I would expect red to attack on a sunny day so that red incurs fewer losses.

(iv) Repeat the simulations in parts (i) and (ii) for $\lambda = 1.5, 2.0, 4.0, 5.0$ and tabulate your results as before. Reconsider your conclusions in part (iii). Are they still valid?

```
In [6]: lambdas=[1.5,2.0,3.0,4.0,5.0]
T2=[battle(lambda,w) for lambda in lambdas,w in ws]
```

```
Out[6]: 5x6 Matrix{Tuple{Int64, String, Float64}}:
(59, "red", 3.28006) (35, "red", 3.60113) ... (10, "red", 4.01786)
(68, "red", 2.65811) (39, "red", 3.08899) (10, "red", 3.65031)
(110, "red", 1.22379) (52, "red", 1.92336) (12, "red", 2.81057)
(98, "blue", 0.599858) (95, "blue", 0.288409) (16, "red", 1.69224)
(58, "blue", 0.97158) (43, "blue", 0.864022) (18, "blue", 0.560668)
```

```
In [7]: using Printf
printcell(v,_,_)=@sprintf("%d\n%s\n%g",v[1],v[2],v[3])
pretty_table(T2,formatters=[printcell],
line_breaks=true,
table_format=TextTableFormat(horizontal_lines_at_data_rows=:all),
stubhead_label="λ \ w",
column_labels=ws,row_labels=lambdas)
```

$\lambda \setminus w$	0.1	0.2	0.25	0.5	0.75	0.9
1.5	59 red 3.28006	35 red 3.60113	29 red 3.69186	16 red 3.91297	11 red 3.99453	10 red 4.01786
2.0	68 red 2.65811	39 red 3.08899	32 red 3.21108	17 red 3.50977	12 red 3.61744	10 red 3.65031
3.0	110 red 1.22379	52 red 1.92336	41 red 2.11985	21 red 2.59439	14 red 2.76393	12 red 2.81057
4.0	98 blue 0.599858	95 blue 0.288409	103 red 0.312059	30 red 1.35226	19 red 1.62403	16 red 1.69224
5.0	58 blue 0.97158	43 blue 0.864022	39 blue 0.818065	27 blue 0.654641	21 blue 0.57854	18 blue 0.560668

For $\lambda = 1.5, 2$ and 3 the advantage to red increases as w increases. In these cases the blue commander would expect red to attack on a sunny day.

For $\lambda = 4$ red wins on a sunny day with the least losses so again the blue commander would expect red to attack on a sunny day. However, if the day is rainy then blue wins. Therefore, there is a chance that blue might attack red on a rainy day.

For $\lambda = 5$ then blue always wins. In this case I would not expect red to ever attack. On the other hand, if I were the red commander I would expect blue to attack on a rainy day.

In []: